

```

*****
'* Name      : CAM2_887.BAS                               *
'* Author    : Alan Paton                                 *
'* Notice    : Copyright (c) 2012                         *
'*           : All Rights Reserved                       *
'* Date      : 04/11/2012                                 *
'* Version   : 2.3                                        *
'* Notes     :                                           *
'*           : Rev. 9.9.2012  4.10.2012  29.10.2012     *
*****

'   Aquire Image for Interak as 80 x 48 pixels
'   using Game Boy Camera module and PIC 16F887
'   Alan Paton Sept 2012          Rev.9.9.2012
'
' Camera module connected as follows:
' 9-Pin connector to PIC
' 1 Vdd      +5V          PIC pin
' 2 START    PortC.1  o/p      16
' 3 Sin      PortC.4  o/p      23
' 4 LOAD     PortC.3  o/p      18
' 5 RESET    PortC.2  o/p      17
' 6 Clk      PortC.0  o/p      15
' 7 READ     PortC.7  i/p      26
' 8 Vout     PortA.0          2
'           analogue i/p from Cam
' 9 Gnd      Gnd
'
' Port B = 8-Bit Data out to Interak

DEFINE ADC_BITS 8          '8-bit result
DEFINE ADC_SAMPLEUS 50    '50uS Sample time
ADCON0 = %11000000
ADCON1 = 0                  'PortA is analogue i/p
ANSEL = %00001011
ANSELH = %00000000 'set PortB digial
TRISA = %11111111 'set PortA to all input
                '(port A.0 = PIC pin 2)
TRISB = %00000000 'set all PortB pins to output
TRISC = %10000000 'set PortC (bit 7 is i/p)
TRISD = %00000010 'set PortD (bit 1 is i/p)
TRISE = %00000000 'set PortE all o/p

' define variables
CLK      VAR PortC.0      'Clock to CAM
START    VAR PortC.1      'Start Image sensing...A/High
RESET    VAR PortC.2      'Register reset..A/Low
LOAD     VAR PortC.3      'Load parameter..A/High
SERIAL   VAR PortC.4      'Data to Cam
NC       VAR PortC.5      'not connected
RDY      VAR PortC.6      'to Interak NMI..A/Low
XREAD    VAR PortC.7      'READ signal from Cam - i/p
IMLED    VAR PortD.0      '"Aquiring Image" LED
STCAM    VAR PortD.1      'Start Cam I/P from Interak
REG       VAR BYTE
D1        VAR BYTE
SBIT     VAR D1.7        'X0000000
V1        VAR BYTE

```

```

R          VAR BYTE
J          VAR BYTE
JW        VAR WORD
CAMX      VAR WORD

' *** Reset Camera ***
CAMERA: PAUSE 500           'wait for 0.5 second
        PortB = %11111111  'all high
        PortC = %11000000  'all low except RDY & XREAD
        PortD = %00000011  'all low except IMLED & STCAM
                               '(Image LED is off)

        PortE = %00000000
        CLK = 1            'CLK high
        CLK = 0            'CLK low
        RESET = 0         'Reset low
        CLK = 1
        RESET = 1         'Reset high
        PAUSE 100         '100mS
        CLK = 0

' Load 8 Registers - in order 1,2,3,4,5,6,7,0
        FOR R=0 TO 14 STEP 2 '8 registers
' 3 address bits...
        READ R,D1          '3-bit address (000 to 111)
        FOR J=1 TO 3
        SERIAL = SBIT      'PortC Pin 4 = Bit 7 of D1
        CLK = 1
        D1 = D1 << 1      'shift 1 place left
        CLK = 0
        NEXT J
' followed by 8 data bits...
        READ R+1,D1        '8-bit data
        FOR J=1 TO 7       'D7 to D1
        SERIAL = SBIT
        CLK = 1
        D1 = D1 << 1      'shift left
        CLK = 0
        NEXT J
' last bit (D0)
        SERIAL = SBIT
        CLK = 1
        LOAD = 1
        CLK = 0
        LOAD = 0
        NEXT R             'next register & data
' Acquire Camera Image
' wait for Start signal from Interak (PortD.1=High)
        CLK = 0
WFSS:   PortB = 0
WF2:    IF STCAM=1 THEN
        GOTO WFH
        ENDIF
        GOTO WF2
'wait for Start signal to return low
WFH:    IF STCAM=0 THEN
        GOTO CAM
        ENDIF
        GOTO WFH
CAM:    IMLED=0            'image sensing LED on

```

```
' 80 x (24x2) pixels = 3840 bytes (3.75K)
  START = 1          'send start bit
  CLK = 1
  START = 0
  CLK = 0
' Exposure
LP1:   CLK = 1
      IF XREAD=1 THEN      'wait for Read high
          GOTO RD1
      ENDIF
      CLK = 0
      GOTO LP1
' Start reading Camera data - only part of image is required.
'skip first 20 lines 127 x 20 = 2540
'-----
RD1:   CAMX = 2540
      FOR JW=1 TO CAMX
      CLK = 0
      ADCIN 0,V1          'get A to D value in V1
                          '(Sample time approx 50uS)

      RDY = 1
      RDY = 1
      CLK = 1
      NEXT JW
RD2:   CAMX = 6350
      FOR JW=1 TO CAMX
RD3:   CLK = 0
      ADCIN 0,V1          'get A to D value in V1
      PortB = V1          'Vout data on PortB
      RDY = 0            'interrupt Interak
'
      pauseus 2
      RDY = 1
      CLK = 1
      NEXT JW
      CLK = 0
RD3:   CLK=1
      IF XREAD=0 THEN      'wait for Read low
          GOTO RD4
      ENDIF
      CLK=0
      GOTO RD3
RD4:   CLK=0
      PortB = $FF          'end of data byte
      RDY = 0            'interrupt
      RDY = 1
      IMLED=1            'switch image LED off
      GOTO CAMERA
'store data in EEPROM starting at address zero
DATA $20,$0E,$40,$00,$60,$80,$80,$01,$A0,00,$C0,$01,$E0,$03,00,
$87

      END
```